3.016

Lecture 20: Linear Homogeneous and Heterogeneous ODEs

Reading:
Kreyszig Sections: 1.4, 1.5 (pages19–25, 26–32)

Ordinary Differential Equations from Physical Models

In engineering and physics, modeling physical phenomena is the means by which technological and natural phenomena are understood and predicted. A model is an abstraction of a physical system, often with simplifying assumptions, into a mathematical framework. Every model should be verifiable by an experiment that, to the greatest extent possible, satisfies the approximations that were used to obtain the model.

In the context of modeling, differential equations appear frequently. Learning how to model new and interesting systems is a learned skill—it is best to learn by following a few examples. Grain growth provides some interesting modeling examples that result in first-order ODES.

Grain Growth

In materials science and engineering, a grain usually refers a single element in an ensemble that comprises a polycrystal. In a single phase polycrystal, a grain is a contiguous region of material with the same crystallographic orientation. It is separated from other grains by *grain boundaries* where the crystallographic orientation changes abruptly.

A grain boundary contributes extra free energy to the entire system that is proportional to the grain boundary area. Thus, if the boundary can move to reduce the free energy it will.

Consider simple, uniformly curved, isolated two- and three-dimensional grains.



Figure 20-18: Illustration of a two-dimensional isolated circular grain and a three-dimensional isolated spherical grain. Because there is an extra energy in the system $\Delta G_{2D} = 2\pi R \gamma_{gb}$ and $\Delta G_{3D} = 4\pi R^2 \gamma_{gb}$, there is a driving force to reduce the radius of the grain. A simple model for grain growth is that the velocity (normal to itself) of the grain boundary is $v_{gb} = M_{gb}\gamma_{gb}\kappa$ where M_{gb} is the grain boundary mobility and κ is the mean curvature of the boundary. The normal velocity v_{gb} is towards the center of curvature.

A relevant question is "how fast will a grain change its size assuming that grain boundary migration velocity is proportional to curvature?"

For the two-dimensional case, the rate of change of area can be formulated by considering the following illustration.

3.016 Home

Full Screen

Close

Quit



Figure 20-19: A segment of a grain boundary moving with normal velocity v_n will move a distance $v_n\Delta t$ in a short time Δt. The motion will result in a change of area $-\Delta A$ for the shrinking grain. Each segment, ds, of boundary contributes to the loss of area by $\Delta A = -v_n\Delta t ds$.

Because for a circle, the curvature is the same at each location on the grain boundary, the curvature is uniform and $v_n = M_{gb}\kappa_{gb}\gamma_{gb} = M_{gb}\gamma_{gb}/R$. Thus

$$\frac{dA}{dt} = -M_{gb}\gamma_{gb}\frac{1}{R}2\pi R = -2\pi M_{gb}\gamma_{gb} \tag{20-1}$$

Thus, the area of a circular grain changes at a constant rate, the rate of change of radius is:

$$\frac{dA}{dt} = \frac{d\pi R^2}{dt} = 2\pi R\frac{dR}{dt} = -2\pi M_{gb}\gamma_{gb} \tag{20-2}$$

which is a first-order, separable ODE with solution:

$$R^2(t) - R^2(t=0) = -2M_{gb}\gamma_{gb}t \tag{20-3}$$

For a spherical grain, the change in volume ΔV due to the motion of a surface patch dS in a time Δt is $\Delta V = v_n\Delta t\, dS$. The curvature of a sphere is

$$\kappa_{sphere} = \left(\frac{1}{R} + \frac{1}{R}\right) \tag{20-4}$$

Therefore the velocity of the interface is $v_n = 2M_{gb}\gamma_{gb}/R$. The rate of change of volume due to the contributions of each surface patch is

$$\frac{dV}{dt} = -M_{gb}\gamma_{gb}\frac{2}{R}4\pi R^2 = -8\pi M_{gb}\gamma_{gb}R == -4(6\pi^2)^{1/3}M_{gb}\gamma_{gb}V^{1/3} \quad (20\text{-}5)$$

which can be separated and integrated:

$$V^{2/3}(t) - V^{2/3}(t=0) = -\text{constant}_1 t \quad (20\text{-}6)$$

or

$$R^2(t) - R^2(t=0) = -\text{constant}_2 t \quad (20\text{-}7)$$

which is the same functional form as derived for two-dimensions.

The problem (and result) is more interesting if the grain doesn't have uniform curvature.



Figure 20-20: For a two-dimensional grain with non-uniform curvature, the local normal velocity (assumed to be proportional to local curvature) varies along the grain boundary. Because the motion is in the direction of the center of curvature, the velocity can be such that its motion increases the area of the interior grain for some regions of grain boundary and decreases the area in other regions.

However, it can still be shown that, even for an irregularly shaped two-dimensional grain, $A(t) - A(t=0) = -(\text{const})t$.

Exact Differential Forms

In classical thermodynamics for simple fluids, expressions such as

$$\begin{aligned} dU =& T dS - P dV \\ =& \left(\frac{\partial U}{\partial S}\right)_V dS + \left(\frac{\partial U}{\partial V}\right)_S dV \\ =& \delta q + \delta w \end{aligned} \tag{20-8}$$

represent the differential form of the combined first and second laws of thermodynamics. If $dU = 0$, meaning that the differential Eq. 20-8 is evaluated on a surface for which internal energy is constant, $U(S, V) = \text{const}$, then the above equation becomes a *differential form*

$$0 = \left(\frac{\partial U}{\partial S}\right)_V dS + \left(\frac{\partial U}{\partial V}\right)_S dV \tag{20-9}$$

This equation expresses a relation between changes in S and changes in V that are necessary to remain on the surface $U(S, V) = \text{const}$.

Suppose the situation is turned around and you are given the first-order ODE

$$\frac{dy}{dx} = -\frac{M(x, y)}{N(x, y)} \tag{20-10}$$

which can be written as the differential form

$$0 = M(x, y)dx + N(x, y)dy \tag{20-11}$$

Is there a function $U(x, y) = \text{const}$ or, equivalently, is it possible to find a curve represented by $U(x, y) = \text{const}$?

If such a curve exists then it depends only on one parameter, such as arc-length, and on that curve $dU(x, y) = 0$.

The answer is, "Yes, such a function $U(x, y) = \text{const}$ exists if an only if $M(x, y)$ and $N(x, y)$ satisfy the Maxwell relations"

$$\frac{\partial M(x,y)}{\partial y} = \frac{\partial N(x,y)}{\partial x} \tag{20-12}$$

Then if Eq. 20-12 holds, the differential form Eq. 20-11 is called *an exact differential* and a U exists such that $dU = 0 = M(x, y)dx + N(x, y)dy$.

Integrating Factors and Thermodynamics

For fixed number of moles of ideal gas, the internal energy is a function of the temperature only, $U(T) - U(T_o) = C_V(T - T_o)$. Consider the heat that is transferred to a gas that changes it temperature and volume a very small amount:

$$\begin{aligned} dU =& C_V dT = \delta q + \delta w = \delta q - PdV \\ \delta q &= C_V dT + PdV \end{aligned} \tag{20-13}$$

Can a Heat Function $q(T, V) = \text{constant}$ be found?

To answer this, apply Maxwell's relations.

Homogeneous and Heterogeneous Linear ODES

A linear differential equation is one that does not contain any powers (greater than one) of the function or its derivatives. The most general form is:

$$Q(x)\frac{dy}{dx} + P(x)y = R(x) \tag{20-14}$$

Equation 20-15 can always be reduced to a simpler form by defining $p = P/Q$ and $r = R/Q$:

$$\frac{dy}{dx} + p(x)y = r(x) \tag{20-15}$$

If $r(x) = 0$, Eq. 20-15 is said to be a *homogeneous linear first-order ODE*; otherwise Eq. 20-15 is a *heterogeneous linear first-order ODE.*

The reason that the homogeneous equation is linear is because solutions can superimposed—that is, if $y_1(x)$ and $y_2(x)$ are solutions to Eq. 20-15, then $y_1(x)+y_2(x)$ is also a solution to Eq. 20-15. This is the case if the first derivative and the function are themselves linear. The heterogeneous equation is also called *linear* in this case, but it is important to remember that sums and/or multiples of heterogeneous solutions are also solutions to the heterogeneous equation.

It will be demonstrated below (directly and with a MATHEMATICA® example) that the homogeneous equation has a solution of the form

$$y(x) = \text{const } e^{-\int p(x)dx} \tag{20-16}$$

To show this form directly, the homogeneous equation can be written as

$$\frac{dy}{dx} = -p(x)y$$

Dividing each side through by through by y and integrate:

$$\int \frac{dy}{y} = \log y = -\int p(x)dx + \text{const}$$

which has solution

$$y(x) = \text{const} e^{-\int p(x)dx}$$

For the case of the heterogeneous first-order ODE, A trick (or, an integrating factor which amounts to the same thing) can be employed. Multiply both sides of the heterogeneous equation by $e^{\int p(x)}$:[12]

$$\exp\left[\int_a^x p(z)dz\right]\frac{dy(x)}{dx} + \exp\left[\int_a^x p(z)dz\right]p(x)y(x) = \exp\left[\int_a^x p(z)dz\right]r(x) \tag{20-17}$$

Notice that the left-hand-side can be written as a derivative of a simple expression

$$\exp\left[\int_a^x p(z)dz\right]\frac{dy(x)}{dx} + \exp\left[\int_a^x p(z)dz\right]p(x)y(x) = \frac{d}{dx}\left\{\exp\left[\int_a^x p(z)dz\right]y(x)\right\} \tag{20-18}$$

[12] The statistical definition of entropy is $S(T,V) = k\log\Omega(U(T,V))$ or $\Omega(U(T,V)) = \exp(S/k)$. Entropy plays the role of integrating factor.

therefore

$$\frac{d}{dx}\left\{\exp\left[\int_a^x p(z)dz\right]y(x)\right\} = \exp\left[p(x)\right]r(x) \tag{20-19}$$

which can be integrated and then solved for $y(x)$:

$$y(x) = \exp\left[-\int_a^x p(z)dz\right]\left\{y(x=a) + \int_a^x r(z)\exp\left[\int_a^z p(\eta)d\eta\right]dz\right\} \tag{20-20}$$

3.016 Home

Full Screen

Close

Quit

Solutions to the General Homogeneous Linear First-Order ODE

The form of MATHEMATICA® 's solution for Eq. 20 is demonstrated.

```
DSolve[
 y'[x] + p[x] y[x] == 0,
 y[x], x]
```
1

$$\left\{\left\{y[x] \rightarrow e^{\int_1^x -p[K[1]]\, dK[1]}\, C[1]\right\}\right\}$$

The dummy integration variables (K[1] in the above) and any integration constants (C[1] above) are picked by Mathematica . Mathematica returns the most general form of homogeneous linear first-order solutiion,

```
DSolve[
 y'[x] + (2 x + 1) y[x] == 0,
 y[x], x]
```
2

$$\left\{\left\{y[x] \rightarrow e^{-x-x^2}\, C[1]\right\}\right\}$$

There is an integration constant above, that will take on a specific value if an additional condition (such as an initial condition, or a boundary condition) is specified

```
DSolve[
 {y'[x] + (2 x + 1) y[x] == 0,
  y[0] == 4}, y[x], x]
```
3

$$\left\{\left\{y[x] \rightarrow 4\, e^{-x-x^2}\right\}\right\}$$

1: `DSolve` solves the linear homogeneous equation first-order ODE $dy/dx + p(x)y = 0$. Two variables are introduced in the solution: one is the 'dummy-variable' of the integration in Eq. 20 which MATHEMATICA® introduces in the form `K[N]` and an integration constant which is given the form `C[N]`.

2: Here, a specific $p(x)$ is given, so the dummy variable doesn't appear if $p(\zeta)$ can be integrated symbolically, as in this case for $p(\zeta) = 2x + 1$.

3: Furthermore, if enough boundary conditions are given to solve for the integration constants, then the `C[N]` are not needed either.

notebook (non-evaluated) pdf (evaluated, color) pdf (evaluated, b&w) html (evaluated)

Solutions to the General Heterogeneous Linear First-Order ODE

This demonstrates the use of `DSolve` to find symbolic solutions of heterogeneous linear homogeneous first-order ODEs: Eq. 20-20. We will see how the homogeneous solution is always part of the sum for a heterogeneous solution.

1

```
DSolve[
 y'[x] + p[x] y[x] == r[x],
 y[x], x]
```

2

$$\{\{y[x] \to e^{\int_1^x -p[K[1]]\,dK[1]} C[1] + e^{\int_1^x -p[K[1]]\,dK[1]} \int_1^x e^{-\int_1^{K[2]} -p[K[1]]\,dK[1]} r[K[2]]\, dK[2]\}\}$$

The solution is general—two dummy integration variables and one constant of integration.

```
homsol =
 DSolve[y'[x] - y[x] == 0,
  y[x], x]
hetsol = DSolve[
  y'[x] - y[x] == e^(2 x),
  y[x], x]
```

3

$$\{\{y[x] \to e^x C[1]\}\}$$

$$\{\{y[x] \to e^{2x} + e^x C[1]\}\}$$

1: `DSolve` solves the general linear heterogeneous equation, $dy/dx + p(x)y = r(x)$, to give the form Eq. 20-20. Note how the homogeneous solution (i.e., the part that depends on `C[1]`) is part of the solution.

2: This is an example for a specific case: $p(x) = -1$ and $r(x) = e^{2x}$. The homogeneous solution is displayed alongside to reinforce that it is always part of the solution.

3.016 Home
Full Screen
Close
Quit

Example: The Bernoulli Equation

The linear first-order ODEs always have a closed form solution in terms of integrals. In general non-linear ODEs do not have a general expression for their solution. However, there are some non-linear equations that can be reduced to a linear form; one such case is the Bernoulli equation:

$$\frac{dy}{dx} + p(x)\, y = r(x)\, y^a \tag{20-21}$$

Reduction relies on a clever change-of-variable, let $u(x) = [y(x)]^{1-a}$, then Eq. 20-21 becomes

$$\frac{du}{dx} + (1-a)p(x)\, u = (1-a)\, r(x) \tag{20-22}$$

which is a linear heterogeneous first-order ODE and has a closed-form solution.

However, not all non-linear problems can be converted to a linear form. In these cases, numerical methods are required.

Changing Variables in Symbolic Differential Equations

The Bernoulli equation, Eq. 20-21, is used to demonstrate how to change variables in an ODE.

```
BernoulliEquation =
 y'[x] + p[x] y[x] ==
  r[x] (y[x])^(a)
```
1

```
yRep = u[x]^(1/(1-a))
DyRep = D[yRep, x]
```
2

```
step1 = BernoulliEquation /.
  {y[x] → yRep,
   y'[x] → DyRep}
```
3

```
step2 = PowerExpand[step1]
```
4

```
step3 = Simplify[step2]
```
5

```
BE = Solve[step3, u'[x]]
```
6

```
uprime = u'[x] /. BE
```
7

```
usol =
 u[x] /. DSolve[u'[x] ==
    uprime[[1]], u[x], x]
```
8

```
ysol =
 (usol[[1]])^(1/(1-a))
```
9

```
BernoulliEquation
Simplify[
 p[x] ysol + D[ysol, x]]
```
10

1: The Bernoulli equation is a non-linear first order ODE, but a series of transformations can turn it into an equivalent linear form.

2: Symbols for what will be used as replacements for $y(x)$ and its derivative in *BernoulliEquation* are defined.

3: For `step1`, the symbols are used for a rule-replacement.

4: Using the form with replacements, the assumption that all variables are real is employed by using `PowerExpand`.

5: `Simplify` produces an equation for which the right-hand-side is zero; thus assuming that $u(x)$ is not identically zero, it can be factored out of the equation.

6: Using `Solve` (n.b, not `DSolve`) to find $u'(x)$ reveals the linear form of Bernoulli's equation in terms of the new variable.

7: The rule that is produced by `Solve` is used to extract the symbolic form of $u'(x)$; the symbolic form of $u'(x)$ is assigned to `uprime`.

8: To extract the solution (`usol`), we use the rule produced by `DSolve` on the equation $u'(x) =$ `usol`.

9: The back-transformation is used to find the general solution $y(x)$ to the non-linear form of the Bernoulli equation (`ysol`).

10: The solution, `ysol`, is plugged back into the left-hand-side of the Bernoulli equation and, with `Simplify`, is shown to be $r(x)$`ysol`^{a}.

3.016 Home

Full Screen

Close

Quit

notebook (non-evaluated) pdf (evaluated, color) pdf (evaluated, b&w) html (evaluated)

Numerical Solutions to Non-linear First-Order ODEs

An example of computing the numerical approximation to the solution to a non-linear ODE is presented. The solutions are returned in the forms of a list of replacement rules to `InterpolatingFunction`. An `InterpolatingFunction` is a method to use numerical interpolation to extract an approximation for any point—it works just like a function and can be called on a variable like `InterpolatingFunction[0.2]`. In addition to the interpolation table, the definition specifies the domain over which the interpolation is considered valid.

Mathematica cannot find a direct solution to the following nonlinear ODE

```
DSolve[
 Sin[2 Pi y'[x]^2] ==
  y[x] x, y[x], x]
```
1

NDSolve is a numerical method for finding a solution. An initial condition and the desired range of solution are required.

```
solution = NDSolve[
  {Sin[2 Pi y'[x]^2] ==
    y[x] x, y[0] == 1},
  y, {x, 0, 3.5}]
```
2

```
{{y → InterpolatingFunction[
    {{0., 3.5}}, <>]},
 {y → InterpolatingFunction[
    {{0., 3.5}}, <>]}}
```

```
y[0.5] /. solution
```
3

```
{0.907437, 1.09733 + 0. i}
```

```
y[Pi] /. solution
```
4

```
{0.0524983,
 2.50186 - 0.61067 i}
```

1: This shows that `DSolve` cannot find a symbolic solution to $\sin[2\pi(y')^2] = y(x)x$.

2: Using NDSolve on a non-linear ODE, the solution is returned as a `InterpolatingFunction` replacement list. Note that there is a warning about "inverse functions" being used to find the solution; this is because of the sin-function which is causing Mathematica to assume a particular domain. There may be more solutions than the two that were returned as an `InterpolatingFunction`.

[: 3–4] This demonstrates how the numerical approximation to the non-linear ODE is obtained at particular values of x.

3.016 Home

Full Screen

Close

Quit

3.016

Plotting Numerical Solutions to Non-linear First-Order ODEs

This is an example of how to extract plot-table expressions from the rules for `InterpolationFunctions` that are returned from `NDSolve`.

```
PStyle = {{Red, Thick},
   {Darker[Green], Thick}};
```
1

```
PlotVanilla =
  Plot[Evaluate[y[x] /.
     solution], {x, 0, 3.5},
   PlotStyle → PStyle,
   PlotRange → {0, 2},
   PlotLabel → "Plot"];
```
2

```
PlotReal = Plot[Evaluate[
    Re[y[x] /. solution]],
   {x, 0, 3.5}, PlotStyle →
    PStyle, PlotLabel →
    "Real Part"];
```
3

```
PlotIm = Plot[Evaluate[
    Im[y[x] /. solution]],
   {x, 0, 3.5}, PlotStyle →
    PStyle, PlotLabel →
    "Imaginary Part"];
```
4

```
GraphicsRow[{PlotVanilla,
  PlotReal, PlotIm},
 ImageSize → Small]
```
5

1: Because `solution` obtained above is a list containing two rules, two curves will be plotted. Here we define a short-hand for the expression that will be passed to `PlotStyle` in the plots below. The first curve will be red, and the second will be `Darker` green.

2: Here, `Plot` is called on the `y[x]` with replacements defined the rule-set for `InterpolatingFunctions`, `solution`, that was obtained from `NDSolve` previously. Using `Evaluate` here immediately creates a list of length two, and plot recognizes this as two curves to which the `PlotStyles` can be applied. If `Evaluate` were not used, then both curves would be be red.

`Plot` only produces curves where the numerical value can be represented by a real number; if a solution has a point where it transforms from real to complex, `Plot` will show a curve that appears to end.

3–4: To determine the solution behavior, the real and imaginary parts are extracted with `Re` and `Im`.

5: This `GraphicsRow` indicates the solution behavior: the first solution is real over the domain where the interpolation is valid; the second solution transforms from real to complex near $x = 0.8$.

3.016 Home
◀◀ ◀ ▶ ▶▶
Full Screen
Close
Quit

Index

3.016 Home

Full Screen

Close

Quit